数式の簡単化と式の書換え

Mapleには、様々な変数や三角関数、特殊関数などを含む数式を簡単にするための機能が豊富に用意されています。この資料ではMapleに用意されている数式簡単化のための基本的なコマンドに加えて、長い数式の別変数への置換えについても紹介します。

簡単化のための基本コマンド

> *restart*

simplify コマンド

基本的な使い方

simplify コマンドは、Maple が内部で持っている数式データベースを参照しながら式を簡単にするための機能を提供しています。もっとも代表的な例が次の三角関数の簡単化です。

> $eq1 := \sin(x)^2 + \cos(x)^2$

$$eq1 := \sin(x)^2 + \cos(x)^2 \quad \textbf{(1.1.1.1)}$$

式 eq1 に simplify コマンドを適用すると、1 という結果を得ます。つまり、$\sin(x)^2 + \cos(x)^2 = 1$ を Maple 上で実現しています。

> $simplify(eq1)$

$$1 \quad \textbf{(1.1.1.2)}$$

また、simplify コマンドには簡単化の目的に応じた色々なオプションを指定することも可能です。次の式 eq を用いて考えます。

> $eq2 := -3\sin(x)^{\frac{1}{2}}\cos(x)^2\sin(x)^m + 3\sin(x)^{\frac{1}{2}}\cos(x)^2\cos(x)^n + 4\sin(x)^{\frac{1}{2}}\cos(x)^4\sin(x)^m - 4\sin(x)^{\frac{1}{2}}\cos(x)^4\cos(x)^n$

$$eq2 := -3\sqrt{\sin(x)}\cos(x)^2\sin(x)^m + 3\sqrt{\sin(x)}\cos(x)^2\cos(x)^n + 4\sqrt{\sin(x)}\cos(x)^4\sin(x)^m - 4\sqrt{\sin(x)}\cos(x)^4\cos(x)^n \quad \textbf{(1.1.1.3)}$$

この式 eq に simplify のみを適用した場合、次の結果です。

> $simplify(eq2)$

$$\sqrt{\sin(x)}\cos(x)^2\left(-3\sin(x)^m + 3\cos(x)^n + 4\cos(x)^2\sin(x)^m - 4\cos(x)^{2+n}\right) \quad \textbf{(1.1.1.4)}$$

しかし、simplify コマンドに size オプションを付け加えて実行すると、より項の個数の少ない結果が得られます。

> $simplify(eq2, size)$

$$-4\cos(x)^2\sqrt{\sin(x)}\left(-\sin(x)^m + \cos(x)^n\right)\left(-\frac{3}{4} + \cos(x)^2\right) \quad \textbf{(1.1.1.5)}$$

Maple は、simplify コマンドに size オプションを指定されると、$\sin(x)^2 + \cos(x)^2 = 1$ のように各種の三角関数の公式や式の変形を伴って項の個数が最小になる結果を探し出します。

ユーザが指定するルールを適用した簡単化

Maple の簡単化では、Maple の数式データベースの公式だけではなく、ユーザが特別に指定したルール（等式で与えられる、簡単化のための式の書換え規則）に基づく手法を適用することも可能です。簡単化を適用するために次の式 eq3 を考えます。

> $eq3 := 8s^4t + 15s^2t^3 - 15s^2t + 7t^5 - 14t^3 + 7t$

$$eq3 := 8s^4t + 15s^2t^3 - 15s^2t + 7t^5 - 14t^3 + 7t \quad \textbf{(1.1.2.1)}$$

例えば、$s^2 + t^2 = 1$ という条件（副関係式）を使って式 eq3 がより簡単になるかどうかを試すには、条件式を2番目の引数として simplify コマンドに指定します。

> $simplify(eq3, \{s^2 + t^2 = 1\})$

$$0 \qquad (1.1.2.2)$$

簡単化の結果がなぜ 0 になるかを確認したい場合は、例えば、式に対して因数分解を行うことで判断できます。

> $factor(eq3)$

$$t\,(8\,s^2 - 7 + 7\,t^2)\,(s^2 + t^2 - 1) \qquad (1.1.2.3)$$

上記のように、この数式 eq3 については $s^2 + t^2 - 1$ という式が含まれており、$s^2 + t^2 = 1$ であることから式 eq3 全体として 0 という結果に簡単化されています。

また、Maple の簡単化は、有理式に対しても機能します。例えば、次の式 eq4 はある伝達関数式だとします。

> $eq4 := \frac{s^5 + s + k}{s - k}$

$$eq4 := \frac{s^5 + s + k}{s - k} \qquad (1.1.2.4)$$

この有理式に対して、$k\,s^5 - k - s = 0$ という条件（副関係式）を用いて簡単化するには、次のようにコマンドを記述します。

> $simplify(eq4, \{k\,s^5 - k - s = 0\}, tdeg(s, k))$

$$\frac{k\,s + k^2 + s + k}{k\,s - k^2} \qquad (1.1.2.5)$$

ここで、$tdeg(s, k)$ は、各単項式内の変数 s, k の次数の和（これを全次数と呼びます）の順序を用いて簡単化を行うことを意味しています。（例えば、s^2, $s \cdot t$, t^3 を tdeg の順序で並び替えると t^3, s^2, $s \cdot t$ となります）

異なる順序で簡単化を行うと、上記とは異なる結果が返されます。以下は、変数に対する辞書式順序を用いて簡単化した例です。

> $simplify(eq4, \{k\,s^5 - k - s = 0\}, plex(k, s))$

$$\frac{s^9 + s^5 - s^4}{-2 + s^5} \qquad (1.1.2.6)$$

このように simplify コマンドを用いて簡単化を行う場合は、変数の順序についても注意しなければなりません。

combine コマンド

combine コマンドは、式を結合させるために用います。式の結合には simplify コマンドと同様に Maple が内部で持っている公式データベースの情報が使われます。

例えば次の三角関数を含む数式を考えます。

> $restart$

> $eq5 := 4\sin(x)\cos(x)^2 - \sin(x) + 8\cos(x)^4 - 8\cos(x)^2 + 1 - \sqrt{4\sin(x)\cos(x)^2 - \sin(x)}$

$$eq5 := 4\sin(x)\cos(x)^2 - \sin(x) + 8\cos(x)^4 - 8\cos(x)^2 + 1 - \sqrt{4\sin(x)\cos(x)^2 - \sin(x)} \qquad (1.2.1)$$

この式に combine コマンドを適用します。

> $combine(eq5)$

$$\sin(3\,x) + \cos(4\,x) - \sqrt{\sin(3\,x)} \qquad (1.2.2)$$

combine コマンドは式を簡単にするだけではなく、演算回数を減少させるという目的においても利用できます。

また、ln 関数（対数関数）を含む式にも combine コマンドを用いることができます。ただし、変数を含む対数関数に combine コマンドを用いる場合、その変数が対数の定義に適合していなければなりません。例えば、次の式 eq6 を考えます。式 eq6 には変数 a, x が含まれています。

> $eq6 := a\ln(x) + 3\ln(x) - \ln(1-x) + \frac{1}{2}\ln(1+x)$

$$eq6 := a\ln(x) + 3\ln(x) - \ln(1-x) + \frac{1}{2}\ln(1+x) \quad \textbf{(1.2.3)}$$

式 eq6 に combine コマンドのみを適用しても $\frac{1}{2}\ln(1+x)$ が $\ln\left(\sqrt{1+x}\right)$ に書き換わるだけです。

> $combine(eq6)$

$$a\ln(x) + 3\ln(x) - \ln(1-x) + \ln\left(\sqrt{1+x}\right) \quad \textbf{(1.2.4)}$$

このような場合、combine コマンドを用いる際に、変数への仮定を与えることでさらに結合させることが可能です。

> $combine(eq6)$ assuming $a :: real, x > 0$

$$a\ln(x) - \ln(1-x) + \ln\left(x^3\sqrt{1+x}\right) \quad \textbf{(1.2.5)}$$

さらに、combine コマンドに anything オプションを指定すると、対数に関する等式変形をすべて適用して結合を行えます。

> $combine(eq6, anything)$ assuming $a :: real, x > 0$

$$\ln\left(\sqrt{1+x}\, x^{a+3}\right) - \ln(1-x) \quad \textbf{(1.2.6)}$$

さらに、変数 x の範囲が (0,1) であるという仮定を付加すると式の最終的な結合結果が得られます。

> $combine(eq6, anything)$ assuming $a :: real, x > 0$ and $x < 1$

$$\ln\left(\frac{\sqrt{1+x}\, x^{a+3}}{1-x}\right) \quad \textbf{(1.2.7)}$$

>

長い数式における別変数を用いた置換え

数式処理の特徴は、面倒な数式計算を自動で行えることです。しかし一方で、計算結果の数式は非常に膨大・長大になることもあります。そのような場合に、Maple の LargeExpressions パッケージと式の整理のための collect コマンドを併用すると、数式の見た目の表現を簡単にすることが可能です。

> *restart*

いま、分母・分子がそれぞれ 8 次、5 次の伝達関数があるとします。ここでは randpoly コマンドにより分母・分子の多項式をランダムに生成し、伝達関数式 tf1 を作ります。

> $d1 := randpoly([s, k], degree = 8, dense):$
> $n1 := randpoly([s, l], degree = 5, dense):$

分母 d1, 分子 n1 を用いて、伝達関数 tf1 は以下のようになります；

> $tf1 := \frac{n1}{d1}$

$$tf1 := (72 - 47 s^5 + 40 s^4 l - 81 s^4 + 91 s^3 l^2 + 68 s^3 l - 10 s^3 + 31 s^2 l^3 - 51 s^2 l^2 + 77 s^2 l + 95 s^2 + s l^4 + s l^3 + 55 s l^2 - 28 s l + 16 s + 30 l^5 - 27 l^4 - 15 l^3 - 59 l^2 - 96 l) / (-49 + 11 k + 29 s + 44 s k + 87 s k^2 - 23 s k^3 + 72 s k^5 + 74 s k^6 + 6 s k^7 + 75 s^2 k + 22 s^7 k - 94 s^6 k^2 + 87 s^6 k - 62 s^5 k^2 + 97 s^5 k - 4 s^4 k^4 - 83 s^4 k^3 - 10 s^4 k^2 + 62 s^4 k + 80 s^3 k^5 - 44 s^3 k^4 + 71 s^3 k^3 - 17 s^3 k^2 - 75 s^3 k - 7 s^2 k^6 - 40 s^2 k^5 - 55 s^7 + 95 k^2 - 56 s^6 - 50 s^2 k^3 - 29 k^3 - 73 s^5 - 8 k^4 - 82 s^4 - 61 k^5 - 10 s^3 + 10 k^6 - 92 s^2 - 23 k^7 + 98 k^8 + 42 s^2 k^4 + 37 s k^4 - 7 s^8 + 23 s^2 k^2) \quad \textbf{(2.1)}$$

式の項数が多いため、非常に見にくい表現となっています。この伝達関数の見た目の表現をシンプルにするには、*s* 以外の係数については別の変数で置き換える（各 *s* の次数の係数部分の多項式を別変数で置き換える）ことを考えます。

この目的のために、Maple の LargeExpressions パッケージを呼び出します。

> *with*(*LargeExpressions*)

$$[LastUsed, Unveil, Veil] \quad \textbf{(2.2)}$$

このパッケージには３つのコマンドが用意されています。ただし、これらのコマンドは単体で用いるのではなく、式の整理を行う collect コマンドと共に用います。Veil (ヴェール）とは隠すという意味です。

伝達関数式 tf1 を、変数 *s* について整理します。その際、collect コマンドの３番目に LargeExpressions パッケージで読み込んだ Veil コマンドを３番目の引数で指定します。*Veil*[*M*] は、*s* の係数多項式を変数 *M* の配列で置き換える、という指定です。

> $collect(tf1, s, Veil[M])$

$$\frac{-47 s^5 + M_1 s^4 + M_2 s^3 + M_3 s^2 + M_4 s + M_5}{-7 s^8 + 11 M_6 s^7 - M_7 s^6 - M_8 s^5 - M_9 s^4 + M_{10} s^3 - M_{11} s^2 + M_{12} s + M_{13}} \quad \textbf{(2.3)}$$

元の tf1 の数式表現に比べてとてもシンプルで見通しがつきやすくなります。

Veil コマンドによって置き換えられた各 M_i には、それぞれの係数（多項式）が割り当てられています。確認するには seq コマンドと Unveil コマンドにより以下のように記述します。Unveil コマンドは Veil によって置き換えられた変数の中身を参照するためのコマンドです。

> $seq(M_i = Unveil[M](M_i), i = 1..15)$

$$M_1 = 40 l - 81, M_2 = 68 l - 10 + 91 l^2, M_3 = 95 - 51 l^2 + 77 l + 31 l^3, M_4 = -28 l + 16 + l^3 + 55 l^2 + l^4, M_5 = 72 - 15 l^3 - 59 l^2 + 30 l^5 - 27 l^4 - 96 l, M_6 = 2 k - 5, M_7 = 94 k^2 - 87 k + 56, M_8 = 62 k^2 - 97 k + 73, M_9 = 83 k^3 - 62 k + 4 k^4 + 82 + 10 k^2, \quad \textbf{(2.4)}$$

$$M_{10} = -17\,k^2 - 44\,k^4 + 71\,k^3 - 75\,k + 80\,k^5 - 10,\ M_{11} = 50\,k^3 - 23\,k^2 + 7\,k^6 + 92 - 75\,k - 42\,k^4 + 40\,k^5,\ M_{12} = 44\,k + 74\,k^6 - 23\,k^3 + 37\,k^4 + 87\,k^2 + 29 + 72\,k^5 + 6\,k^7,\ M_{13} = -49 + 11\,k - 8\,k^4 - 29\,k^3 + 98\,k^8 + 10\,k^6 - 61\,k^5 + 95\,k^2 - 23\,k^7,\ M_{14} = M_{14},\ M_{15} = M_{15}$$

Veil コマンド・collect コマンドを使うと、要素が多項式となっている行列でも微分方程式等でも見た目の表現を整理することが可能です。

v, w を変数とするランダムな多項式を要素に持つ次の行列 G を定義します。行列であるにも関わらず非常に見にくい表現です。

> $G := Matrix(4, 4, (i, j) \rightarrow randpoly([v, w], degree = rand(1..2)()))$

$$G := [[47\,v - 90\,w + 43, -91 - 88\,v - 48\,w + 53\,v^2 - 28\,v\,w + 5\,w^2, -10\,v - 82\,w + 71, 83 + 9\,v - 60\,w - 83\,v^2 + 98\,v\,w - 48\,w^2], [62\,v + 37\,w + 5, -17 + 25\,v + 91\,w + 98\,v\,w - 64\,w^2, -90 - 60\,v - 34\,w - 13\,v^2 + 44\,v\,w - 2\,w^2, -47\,v - 39\,w - 53], [-97 + 33\,v + 10\,w + 7\,v^2 - 89\,v\,w + 65\,w^2, -25 - 96\,v + 50\,w - 60\,v^2 - 42\,v\,w + 7\,w^2, -70\,v + 34\,w - 68, 16 + 52\,v - 20\,w - 4\,v^2 - 89\,v\,w - 77\,w^2], [80\,v + 28\,w - 42, 21\,v - 35\,w + 97, -64 + 89\,v - 16\,w + 59\,v^2 - 69\,v\,w - 46\,w^2, 87\,v - 34\,w + 40]] \tag{2.5}$$

map コマンドにより collect コマンドを用いた変数置換えをすべての要素に適用します。

> $map(p \rightarrow collect(p, v, Veil[W]), G)$

$$[[47\,v - W_{24}, 53\,v^2 - 4\,W_{18}\,v + W_{19}, -10\,v - W_{12}, -83\,v^2 + W_5\,v - W_6], [62\,v + W_{23}, W_{16}\,v - W_{17}, -13\,v^2 + 4\,W_{10}\,v - 2\,W_{11}, -47\,v - W_4], [7\,v^2 - W_{21}\,v + W_{22}, -60\,v^2 - 6\,W_{14}\,v + W_{15}, -70\,v + 34\,W_9, -4\,v^2 - W_2\,v - W_3], [80\,v + 14\,W_{20}, 21\,v - W_{13}, 59\,v^2 - W_7\,v - 2\,W_8, 87\,v - 2\,W_1]] \tag{2.6}$$

大文字の W で置き換えられた中身を確認するには、上と同様に Unveil コマンドを用います。LastUsed コマンドは変数で用いたインデックスの最後の番号を返します。

> **for** i **from** 1 **to** $LastUsed[W]$ **do**
> $W_i = Unveil[W](W_i)$
> **end do**

$$W_1 = 17\,w - 20$$

$$W_2 = 89\,w - 52$$

$$W_3 = -16 + 77\,w^2 + 20\,w$$

$$W_4 = 39\,w + 53$$

$$W_5 = 98\,w + 9$$

$$W_6 = -83 + 48\,w^2 + 60\,w$$

$$W_7 = 69\,w - 89$$

$$W_8 = 32 + 23\,w^2 + 8\,w$$

$$W_9 = w - 2$$

$$W_{10} = 11\,w - 15$$

$$W_{11} = 45 + w^2 + 17\,w$$
$$W_{12} = 82\,w - 71$$
$$W_{13} = 35\,w - 97$$
$$W_{14} = 7\,w + 16$$
$$W_{15} = -25 + 7\,w^2 + 50\,w$$
$$W_{16} = 25 + 98\,w$$
$$W_{17} = 17 - 91\,w + 64\,w^2$$
$$W_{18} = 7\,w + 22$$
$$W_{19} = -91 + 5\,w^2 - 48\,w$$
$$W_{20} = 2\,w - 3$$
$$W_{21} = 89\,w - 33$$
$$W_{22} = -97 + 65\,w^2 + 10\,w$$
$$W_{23} = 37\,w + 5$$
$$W_{24} = 90\,w - 43 \qquad \textbf{(2.7)}$$

Maple に用意されている簡単化や式の整理のためのコマンドを用いることで、手計算を必要とせずに式整理を効率的に行え、また簡単な計算間違いも軽減できます。